*Use TACS to replace IOCS estimates of office/street split, 
 and for control totals by craft subgroup, route group, MODS group;

libname libIOCS 'C:\IOCSData' COMPRESS=YES;
option mprint symbolgen;
%let tal14 = libIOCS.tally14;
%let tal13 = libIOCS.tally13v3;

data tally13; set &tal13.;
  where craft = 'CARR' and F261 ne '4';
  *define craft subgroup;
  CraftGrp = F9252*1;
  select (CraftGrp);
	when (5)	CraftGrpDesc = "Reg_FT";
	when (6)	CraftGrpDesc = "CCA_PT";
	otherwise CraftGrpDesc = "Unk   ";
  end; *select;
  *define route group;
  if '71' <= F260 <= '83' then routeGrp = 'REG';
  else if '86' <= F260 <= '98' then routeGrp = 'SPR';
  else if  F260 = '99' then routeGrp = 'MIX';
  else routeGrp = 'UNK';
  *define MODS group;
  select (Q16A01);
  	when ('A') MODSType = "OFFICE";
  	when ('B') MODSType = "STREET";
  	otherwise  MODSType = "UNKN  ";
  end; *select;
run;

*check for problem routes;
data tmpProb; set tally13;
where CraftGrp in (5,6) and RouteGrp = 'UNK';
run;
proc print data=tmpProb;title "Problem routes"; run; title;

*Simulate getting total costs from trial balance by craft subgroup;
%let grp=CraftGrp;
proc sql;
create table totCarrCraftGrpCosts as
  select &grp., sum(x9250) as CostK format dollar12.0
  from tally13 
  where CraftGrp in (5,6) /*and RouteGrp in ('REG','SPR')*/
  group by &grp.;
quit;

*Assign route group for TACS hours;
data CarrHrsDat; set libIOCS.TACSCarrSummHrsFY13;
  select (LDC);
	when (2000)      			RouteGrp = "SUP";
	when (2100,2200,2800,2900) 	RouteGrp = "REG";
	when (2300,2600,2700) 		RouteGrp = "SPR";
	when (2500) 				RouteGrp = "RUR";
	otherwise 					RouteGrp = "UNK";
  end; *select;
  if CraftGrp = 1 then CraftGrp = 5; *convert supervisors to fulltime carriers;
  select (CraftGrp);
	when (1,5)	CraftGrpDesc = "Reg_FT";
	when (6)	CraftGrpDesc = "CCA_PT";
	otherwise 	CraftGrpDesc = "Unk   ";
  end; *select;
run;

proc freq data=CarrHrsDat;
table MODSType CraftGrp CraftGrp*MODSType;
weight workhrs;
where MODSType in ("OFFICE","STREET");
run;

%let grp=CraftGrp, CraftGrpDesc, RouteGrp, MODSType;
proc sql;
create table totTACSCarrHrs1 as
  select &grp., sum(workhrs) as workhrs format comma12.0
  from CarrHrsDat /*libIOCS.TACSCarrSummHrsFY13*/
  where CraftGrp in (5,6) and RouteGrp in ('REG','SPR') and MODSType in ('OFFICE','STREET')
  group by &grp.;
create table totTACSCarrHrs2 as
  select CraftGrp, RouteGrp, sum(workhrs) as totWorkhrs format comma12.0
  from totTACSCarrHrs1 
  group by CraftGRp, RouteGrp;
quit;

%let grp5=craftGrp, MODSType;
proc sql;
create table totTACSCarrHrs5 as
  select t1.*, totWorkHrsCrftMODS, workhrs/totWorkHrsCrftMODS as pctCrftMODS format percent9.2
  from totTACSCarrHrs1 as t1 left join  
  	(select &grp5., sum(workhrs) as totWorkHrsCrftMODS format comma12.0
	 from totTACSCarrHrs1 group by &grp5.) as t2
	on t1.craftGrp=t2.craftGrp and t1.MODSType = t2.MODSType
  ;
quit;

proc export data=totTACSCarrHrs5
        outfile = "C:\TempMCZ\tmptotTACSCarrHrs5_FY13"
        DBMS = Excel2007 REPLACE;
run;

proc sql;
  create table tallyMIXRoute as
  	select t.*, h.RouteGrp as RoutePool, pctCrftMODS, x9250*pctCrftMODS as x9250b format dollar12.0
	from tally13 as t left join totTACSCarrHrs5 as h
	on t.CraftGrp = h.CraftGrp and t.MODSType=h.MODSType
	where t.routeGrp = "MIX";
quit;
proc sql;
create table tally13b as
select * from tallyMIXRoute
union
select *, RouteGrp as RoutePool, 1. as pctCrftMODS, x9250 as x9250b 
  from tally13 where routeGrp ne "MIX";
quit;


%let grp=CraftGrp, routePool, MODSType;
proc sql;
create table totCostsByGrp1 as 
  select &grp., sum(x9250b) as CostK format dollar12.0
  from tally13b 
  where CraftGrp in (5,6) 
  group by &grp.;
create table totCostsByGrp2 as
  select t1.*, totCostsCrft, CostK/totCostsCrft as pctCrft format percent9.1
  from totCostsByGrp1 as t1 left join
   (select CraftGrp, sum(CostK) as totCostsCrft format dollar12.0
	from totCostsByGrp1 group by CraftGrp) as t2
   on t1.CraftGrp = t2.CraftGrp;
quit;

proc sql;
create table totTACSCarrHrs4 as
  select t1.*, totWorkhrs, workhrs/totWorkhrs as pctHours format percent7.1 
  from totTACSCarrHrs1 as t1 left join 
	(select CraftGrp, sum(workhrs) as totWorkhrs format comma12.0
	 from totTACSCarrHrs1 group by CraftGrp) as t2
  on t1.CraftGrp = t2.CraftGrp;
quit;
/*
proc sql;
create table totTACSCarrHrs3 as
  select t.*, CostK as totCostK
  from totTACSCarrHrs2 as t left join TotCarrCraftGrpCosts as c
    on t.CraftGrp = c.CraftGrp ;
create table totTACSCarrHrs4 as
  select t1.*, totWorkHrs, workHrs/totWorkHrs as pctHours format percent7.1, 
  	totCostK, totCostK*workHrs/totWorkHrs as TACSCostK format dollar12.0
  from totTACSCarrHrs1 as t1 left join totTACSCarrHrs3 as t3
    on t1.CraftGrp = t3.CraftGrp and t1.RouteGrp = t3.RouteGrp;
quit;
*/

*merge IOCS with TACS to get adjustment factors;
proc sql;
create table IOCSTACSFactors as
  select c.*, h.*, totWorkHrs, workHrs/totWorkHrs as pctHours format percent8.1,
	totCostsCrft*workHrs/(CostK*totWorkHrs) as TACSFactor
    from totCostsByGrp2 as c left join totTACSCarrHrs4 as h
	on c.CraftGrp = h.CraftGrp and c.RoutePool=h.RouteGrp and c.MODSType=h.MODSType;
quit;
proc export data=IOCSTACSFACtors
        outfile = "C:\TempMCZ\tmpIOCSTACSFactors"
        DBMS = Excel2007 REPLACE;
run;


*reweight tallies using TACS: use IOCSTACSFactors:
	- use new tally weights, x9250b, that have split MIX routes to SPR and REG route pool
	- REG and SPR (RoutePool) tally weights are using by TACS office/street percentage for REG and SPR, by craft subgroup
	- all clocked-to-street tallies are zeroed out, even if on premises;
*	CostKTACS1 would be adjusted by TACS and add up to original x9250, 
		but now reduced by cost for MIX tallies in-office, so only adds up to REG and SPR;
*	CostKTACS2 is CostKTACS1 with zero weight for street tallies;
*	CostKTACS3 is same as CostKTACS4, but street tallies not zeroed out, 
		should add up to total of x9250;
proc sql;
create table tally13c as
  select t.*, TACSFactor, x9250b*TACSFactor as CostKTACS1 format dollar12.0,
  	case (t.MODSType)
		when "OFFICE" then  x9250b*TACSFactor
		when "STREET" then 	0
		else 0 
	end as CostKTACS2 format dollar12.0
  from tally13b as t left join IOCSTACSFactors as f
  	on t.CraftGrp=f.CraftGrp and t.routePool=f.routeGrp and t.MODSType=f.MODSType;
quit;


%let grp1=IOCSProdGrp, IOCSProdCode, f262;
%let grp2=IOCSProdGrp, IOCSProdCode;
%let grp3=CRAClass;
%let grp4=RouteGrp,MODSType;*,Q16F02A;
%let grp5=RouteGrp,MODSType,f262; *Q16F02A;
proc sql;
create table totIOCS1 as
  select &grp1., sum(x9250b) as CostK format dollar12.0, 
	sum(CostKTACS1) as CostKTACS1 format dollar12.0,
	sum(CostKTACS2) as CostKTACS2 format dollar12.0
  from tally13c
  group by &grp1.;
create table totIOCS2 as
  select &grp2., sum(x9250b) as CostK format dollar12.0, 
	sum(CostKTACS1) as CostKTACS1 format dollar12.0,
	sum(CostKTACS2) as CostKTACS2 format dollar12.0
  from tally13c
  group by &grp2.;
create table totIOCS3 as
  select &grp3., sum(x9250b) as CostK format dollar12.0, 
	sum(CostKTACS1) as CostKTACS1 format dollar12.0,
	sum(CostKTACS2) as CostKTACS2 format dollar12.0
  from tally13c
  group by &grp3.;
create table totIOCS4 as
  select &grp4., sum(x9250b) as CostK format dollar12.0, 
	sum(CostKTACS1) as CostKTACS1 format dollar12.0,
	sum(CostKTACS2) as CostKTACS2 format dollar12.0
  from tally13c
  group by &grp4.;
create table totIOCS5 as
  select &grp5., sum(x9250b) as CostK format dollar12.0, 
	sum(CostKTACS1) as CostKTACS1 format dollar12.0,
	sum(CostKTACS2) as CostKTACS2 format dollar12.0
  from tally13c
  where f262 in ('6430','6519','6522','6422')
  group by &grp5.;
quit;
data totIOCS1b; set totIOCS1;
  PctChange = CostKTACS/CostK-1; 
  format PctChange percent7.1;
data totIOCS2b; set totIOCS2;
  PctChange = CostKTACS/CostK-1;
  format PctChange percent7.1;
run;
data totIOCS3b; set totIOCS3;
  PctChange1 = CostKTACS1/CostK-1;
  PctChange2 = CostKTACS2/CostK-1;
  format PctChange1 PctChange2 percent7.1;
run;
proc export data=totIOCS3b
        outfile = "C:\TempMCZ\tmptotIOCS3b"
        DBMS = Excel2007 REPLACE;
run;
proc export data=totIOCS4
        outfile = "C:\TempMCZ\tmptotIOCS4"
        DBMS = Excel2007 REPLACE;
run;
proc export data=totIOCS5
        outfile = "C:\TempMCZ\tmptotIOCS5"
        DBMS = Excel2007 REPLACE;
run;



